

Gramática corrigida da linguagem Pyscal

Programa → Classe \$

Classe → "class" ID ":" ListaFuncao Main "end" "."

DeclararID → TipoMacro ID ";"

ListaFuncao → ListaFuncao'

ListaFuncao' → Funcao ListaFuncao' | ε

Funcao → "def" TipoMacro ID "(" ListaArg ")" ":" FKDeclararID ListaCmd Retorno "end" ";"

FKDeclararID → DeclararID FKDeclararID | ε

ListaArg → Arg ListaArg' | ε

ListaArg' → "," Lista Arg | ε

Arg → TipoMacro ID

Retorno → "return" Expressao ";" | ε

Main → "defstatic" "void" "main" "(" "String" "[" "]" ID ")" ":" FKDeclararID ListaCmd "end" ";"

TipoMacro → TipoPrimitivo TipoMacro'

TipoMacro' → "[" "]" | ε

TipoPrimitivo → "bool" | "integer" | "String" | "double" | "void"

ListaCmd → ListaCmd'

ListaCmd' → Cmd ListaCmd' | ε

Cmd → CmdIF | CmdWhile | CmdAtribui | CmdFuncao | CmdWrite | CmdWriteln

CmdIF → "if" "(" Expressao ")" ":" ListaCmd "CmdIF'

CmdIF' → "end" ";" | "else" ":" ListaCmd "end" ";"

CmdWhile → "while" "(" Expressao ")" ":" ListaCmd "end" ";"

CmdWrite → "write" "(" Expressao ")" ";"

CmdWriteln → "writeln" "(" Expressao ")" ";"

CmdAtribui → ID CmdAtribui'

CmdAtribui' → "=" Expressao ";" | "[" Expressao "]" "=" Expressao ";"

CmdFuncao → ID "(" QBExp ")" ";"

FKExp → "," Expressao FKExp | ε

QBExp → Expressao FKExp | ε

Expressao → Expressao1 Expressao'

Expressao' → and Expressao1 Expressao' | or Expressao1 Expressao' | ε

Expressao1 → Expressao2 Expressao1'

Expressao1' → > Expressao2 Expressao1' | >= Expressao2 Expressao1' | < Expressao2 Expressao1' | <= Expressao2 Expressao1' | == Expressao2 Expressao1' | != Expressao2 Expressao1' | ε

Expressao2 → Expressao3 Expressao2'

Expressao2' → + Expressao3 Expressao2' | - Expressao3 Expressao2' | ε

Expressao3 → Expressao4 Expressao3'

Expressao3' → * Expressao4 Expressao3' | / Expressao4 Expressao3' | ε

Expressao4 → ID Expressao4' | ConstInteger | ConstDouble | ConstString | "true" | "false" | "vector" TipoPrimitivo "[" Expressao "]" | OpUnario Expressao | "(" Expressao")"

Expressao4' → "[" Expressao "]" | "(" QBExp ")" | ε

OpUnario → "-" | "!"